
Emic Text-To-Speech Module (#30006)

General Description

The Emic Text-to-Speech (TTS) Platform is a modular hardware design for text-to-speech applications. The Emic 
Platform is a fully integrated module that converts a stream of digital text into a high-quality English-speaking voice. It 
provides a simple way to speech-enable any embedded application. Requiring a simple 2-wire serial interface and 
two optional I/O lines for status notification, the Emic Platform can easily be integrated into nearly any new or existing 
design. 

The Emic Platform is controlled through a simple set of commands and off-loads most of the computational overhead 
required for TTS from the host controller. The Emic Platform command set consists of a number of functions for core 
TTS functionality and abstracts the low-level hardware from the end-user application, making it extremely simple to 
add TTS capabilities to any product.

Features

 Fully-integrated, high-quality speech synthesis on a single module
 Elegant platform provides for unlimited text-to-speech applications
 Simple interface to BASIC Stamp® and Javelin Stamp™

Module Highlights

 Serial TTL interface (2-wire, 2400 baud)
 Requires single +5VDC supply
 Compact size: 2.0" L x 1.375" W
 On-board 8 Ω, 300 mW speaker driver
 Easy-to-use ASCII or hexadecimal command sequences
 Bi-color LED for visual indication of activity
 0.100" pin spacing for easy prototyping and integration
 -40 ºC to +85 ºC operating temperature

Pin Description

Pin 
(OEM1)

Pin
(SIP)

Pin Name Type Function

1 7 VCC P System power, +5V DC input. 
2 3 BUSY O Busy notification line. Digital logic HIGH when Emic Platform is in-use 

(e.g., receiving characters, converting text-to-speech, etc.) or when a 
system error has occurred. Logic LOW when the Emic Platform is idle 
and ready to receive new commands or text. The visible LED on the 
Emic Platform is red when BUSY is HIGH and green when BUSY is 
LOW.

3 1 SIN I Serial In. TTL-level interface, 2400bps, 8 data bits, no parity, 1 stop bit.
4 2 SOUT O Serial Out. TTL-level interface, 2400bps, 8 data bits, no parity, 1 stop 

bit.
5 8, 11 GND G System ground. Connect to power supply’s ground (GND) terminal.
6 14 /RESET I External system reset pin. Active LOW digital input. Bring this pin LOW 

to reset the Emic Platform. Recommended use is after the host 
controller power-up or an Emic Platform system error. This pin should 
be left floating if it is not used. 

7 10 AOUT 
(ANA)

O Audio output buffer pin provides a single-ended analog output of the 
synthesized speech for external amplification. Drives a minimum load of 
5k Ω up to a maximum of 1Vp-p. The AC signal is superimposed on 
approximately 1.2VDC bias and must be capacitively coupled to the 
load. 

8 15 SP- O Differential negative speaker driver output. Designed to drive an 8 Ω 
speaker up to a maximum of 23.5mW for the OEM version and 300mW 
for the SIP version. Do NOT connect this pin if using SP+ as single-
ended output (OEM version only). 

9 16 SP+ O Differential positive speaker driver output. Designed to drive an 8 Ω 
speaker up to a maximum of 23.5mW for the OEM version and 300mW 
for the SIP version. On the OEM version, this pin is biased to 
approximately 1.2VDC and can be used as a single-ended output if 
capacitively coupled to the load. 

--- 9 AIN I Analog input pin for SIP version only. Allows an audio signal to be fed 
through the Emic Platform to either the AOUT or SP+/SP- pins. The 
input has a nominal 694mVp-p level. There is no gain control on this 
signal. The audio signal must be AC coupled to the AIN pin. No signal 
should be active on the AIN pin when it is not in use, or else the signal 
may bleed through to the Emic Platform's audio output at undesirable 
times. This pin should be left floating if it is not implemented. 

1. For information only; the OEM version is no longer in production.

Note: Type: I = Input, O = Output, I/O = Bidirectional, P = Power, G = Ground

NOTE: All unspecified pins should be left floating. They should not be connected at any time. Connecting unspecified 
pins to any signal, ground, or VCC may result in improper device behavior or cause damage to the device.  

Block Diagram

Note: The AIN pin is not available on the OEM version.

Functional Description

Overview

The Emic Platform is a fully integrated module that converts a stream of digital text into a high-quality English-
speaking voice. The Emic Platform performs text-to-speech synthesis and host processor communications. The 
system architecture consists of the following functions:

 Serial interface to interpret commands and data using a simple command structure and menuing system
 Text normalization module to pre-process incoming text into pronounceable words
 Word-to-phoneme translator which converts incoming words into their phonetic representation
 Phoneme mapping to match incoming phonemes into their corresponding pre-recorded audio sample
 Volume, pitch, and speed adjustments
 On-board audio amplifier (300 mW)

The Emic Platform ships with a standard American English-speaking female voice with a San Francisco Bay Area 
dialect. The Emic Platform is designed to support different languages or voices as they become available.  

Physical Connections

The Emic Platform can be integrated into any design using a minimum of five connections (VCC, GND, SIN, SOUT, 
and AOUT). Optional status and control pins are also available if desired. Though many configurations are possible, a 
typical connection diagram to a microprocessor uses eight connections (VCC, GND, SIN, SOUT, BUSY, /RESET, 
SP-, SP+).

Basic Connection Diagram: Emic Platform with Microprocessor Host Control

   Cm

GND

VCC

5V 5V

Emic TTS
Platform

GND

AINBUSY

/RESET AOUT

Communication Protocol

The Emic Platform is controlled by the host via an easy-to-use, TTL-level, asynchronous, two-wire serial 
communications interface. The SIN (Serial Input) pin receives data and commands from the host controller to the 
Emic Platform. The SOUT (Serial Output) pin transmits data and responses to the host controller from the Emic 
Platform. 

The Emic Platform can connect directly to any TTL-compatible UART or to an RS232-compatible interface by using 
an external level shifter. 

All communication is 8 data bits, no parity, 1 stop bit, non-inverted, least significant bit first (8N1). The baud rate is 
configured in the Emic Platform firmware for 2400 baud, a standard communications speed supported by the BASIC 
Stamp microcontroller, the Javelin Stamp microcontroller, and most any other microcontroller or PC. The host 
controller must initiate all communication. The Emic Platform does not initiate any communication. Refer to the 
section on Command Structure for a list of available commands. 

Configuration Switches

The two-position DIP switch located on the Emic Platform is used to configure the communications properties of the 
device to fit a wide range of applications. 

If SW1 is set to ON, the Emic Platform will accept and respond to commands in an ASCII (printable text) format. If 
SW1 is set to OFF, the Emic Platform will accept and respond to commands in a hexadecimal format. Refer to the 
section on Command Structure for details. 

If SW2 is set to ON, each byte sent to the Emic Platform will be echoed back to the host processor. If SW2 is set to 
OFF, the Emic Platform will not echo each byte to the host processor and will only transmit the command response 
(such as "OK", "ERROR", or other information) This switch is useful for applications where the user needs to ensure 
the data is being properly received by the Emic Platform or to verify that the correct data is being typed, such as 
when the Emic Platform is hooked up to a PC and simple terminal program. 

Control Lines

In addition to the serial communications interface, the Emic Platform has two control lines. The BUSY pin is used by 
the Emic Platform to notify the host controller that it is busy and cannot accept new commands or data (such as 
during a text-to-speech conversion). The BUSY pin is set HIGH when the Emic Platform is in use. The BUSY pin is 
set LOW when the Emic Platform is in the idle state and ready to receive a new command. 

A visual indication of the state of the Emic Platform is given with the on-board LED (Light Emitting Diode). When 
BUSY is HIGH, the LED is red. When BUSY is LOW, the LED is green. On initial power-up and during device reset, 
the on-board LED will be both red and green while the Emic Platform is initializing. 

The /RESET pin is an active LOW digital input used to "hard reset" the Emic Platform, which reinitializes the module's 
control firmware and resets the volume, speed, and pitch settings to their default values. This pin should be left 
floating if it is not used, or else it should remain in a logic HIGH state until a system reset is desired. 

Command Structure

The Emic Platform can accept commands in either an ASCII (printable text) or hexadecimal format depending on the 
configuration DIP switch setting (refer to Configuration Switches section). The hexadecimal format is essentially an 
abbreviated form of the longer ASCII command structure. Both command formats result in the same operation. 

Available Command Set

Command ASCII Hexadecimal
Convert text to speech say=text; 0x00, text, 0xAA
Set volume volume=n; 0x01, n, 0xAA
Set speed speed=n; 0x02, n, 0xAA
Set pitch pitch=n; 0x03, n, 0xAA
Add abbreviation addabbr=abbr,text; 0x04, abbr, 0x2C, text, 0xAA
Delete abbreviation delabbr=abbr; 0x05, abbr, 0xAA
List abbreviations listabbr; 0x06, 0xAA
Emic version numbers version; 0x07, 0xAA
Device soft reset reset; 0x08, 0xAA
Analog audio input (SIP only) audio; 0x09, 0xAA
Emic help help; or ?; 0xFE, 0xAA

The Emic Platform has a 128-byte circular receive buffer, meaning that any combination of command and data must 
be less than 128 bytes in order to be properly handled. If this limit is exceeded, the pointer in the buffer will reset to 
the beginning and overwrite previously received characters. A single word may have a maximum length of 53 
characters. Any words longer than 53 characters will be truncated. 

The Emic Platform will process the buffer when it receives the termination character. In ASCII mode, the termination 
character is a ";" (semicolon) or 0x0D byte (carriage return). In hexadecimal mode, it is a 0xAA byte. In order to clear 
the receive buffer, such as in the case of a typo or error, a single termination character should be sent, depending on 
the ASCII or hexadecimal mode of the Emic Platform. 

In ASCII mode, if a valid command is entered and the operation is successful, the Emic Platform will respond with 
"OK". If an incorrect command or data format is entered or if the operation has failed, the Emic Platform will respond 
with "ERROR". 

In hexadecimal mode, if a valid command is entered and the operation is successful, the Emic Platform returns a 
single 0x55 byte. If an incorrect command or data format is entered or if the operation has failed, the Emic Platform 
returns a single 0xFF byte.

Convert text to speech:

This command begins the text-to-speech conversion process using the passed text string. The audio will be output on 
both the single-ended AOUT pin and the differential SP+/SP- speaker driver output. Refer to the sections on 
Allowable ASCII Characters and Phonetic Alphabet for more details. 

During this operation, the visible LED will remain RED and the BUSY line will be set HIGH until all passed text has 
been converted to speech. Once the text-to-speech conversion begins, it will continue until the process is complete. 
There is no way to interrupt the process, other than pulling the /RESET pin LOW to reinitialize the Emic Platform. 
When the text-to-speech conversion is complete, the LED will return to GREEN and the BUSY line will be set LOW.

Command example: Convert the text "hello" to audible speech

ASCII Hexadecimal
To Emic Platform say=hello; 0x00, 0x68, 0x65, 0x6C, 0x6C, 0x6F, 

0xAA
From Emic Platform OK 0x55

Set volume:

This command sets the audio output volume of the Emic Platform. A single byte, n, is passed along with the 
command to specify the desired volume. A digitally controlled attenuator on the Emic Platform provides volume 
control in 8 steps of 4 dB, referenced to the output. Total volume range is from -28 dB (softest, n = 0) to 0 dB 
(loudest, n = 7). Default volume is set to -12 dB (n = 4).

In addition to the range of decimal numbers, "+" or "-" can also be passed to increment or decrement the current 
volume by 1 step. This will have no effect if an increment command is sent when the volume is already at the loudest 
setting or if a decrement command is sent when the volume is already at the softest setting. When the Emic Platform 
is reset or when power is removed, the volume setting will revert to its default value. 

Command example: Set volume to n = 3 (-16dB)

ASCII Hexadecimal
To Emic Platform volume=3; 0x01, 0x33, 0xAA
From Emic Platform OK 0x55

Set speed:

This command sets the speed at which the converted speech is spoken. A single byte, n, is passed along with the 
command to specify the desired speed. Total speed range is from n = 0 (slowest) to n = 4 (fastest). Default speed is 
set to n = 2. 

In addition to the range of decimal numbers, "+" or "-" can also be passed to increment or decrement the current 
speed by 1 step. This will have no effect if an increment command is sent when the speed is already at the fastest 
setting or if a decrement command is sent when the speed is already at the slowest.

When using concatenative speech synthesis systems such as the Emic Platform, speech speed and pitch are often 
adjusted together to maintain an audible and understandable output. However, it is not required, and each may be 
adjusted separately if desired. When the Emic Platform is reset or when power is removed, the speed setting will 
revert to its default value. 

Command example: Increment speed setting by one step

ASCII Hexadecimal
To Emic Platform speed=+; 0x02, 0x2B, 0xAA
From Emic Platform OK 0x55

Set pitch:

This command sets the pitch of the converted speech. A single byte, n, is passed along with the command to specify 
the desired pitch. Total pitch range is from n = 0 (lowest) to n = 6 (highest). Default pitch is set to n = 1. 

In addition to the range of decimal numbers, "+" or "-" can also be passed to increment or decrement the current pitch 
by 1 step. This will have no effect if an increment command is sent when the pitch is already at the highest setting or 
if a decrement command is sent when the pitch is already at the lowest.

When using concatenative speech synthesis systems such as the Emic Platform, speech speed and pitch are often 
adjusted together to maintain an audible and understandable output. However, it is not required, and each may be 
adjusted separately if desired. When the Emic Platform is reset or when power is removed, the pitch setting will revert 
to its default value. 

Command example: Decrement pitch setting by one step

ASCII Hexadecimal
To Emic Platform pitch=-; 0x03, 0x2D, 0xAA
From Emic Platform OK 0x55

Add abbreviation:

This command adds a user-specified abbreviation to the abbreviation table stored on the Emic Platform. The 
abbreviation, abbr, and its corresponding full word or sentence, text, are passed along with the command. Once an 
abbreviation is loaded into the Emic Platform, the full text will replace the abbreviation any time the abbreviation is 
used during a text-to-speech conversion. 

If multiple abbreviations are added with the same abbreviation string, the Emic Platform will use the first instance. 
The expanded text for the abbreviation cannot contain any numerals (such as "Basic Stamp 2"), only alphabetic 
characters (such as "Basic Stamp Two").

When an abbreviation is successfully added, the Emic Platform performs a "soft reset", which resets the volume, 
speed, and pitch settings to their default values. Refer to the section on Abbreviations for details. 

Command example: Convert SMS/Instant Message abbreviation into understandable word, ":)" into "happy"

ASCII Hexadecimal
To Emic Platform addabbr=:),happy; 0x04, 0x3A, 0x29, 0x2C, 0x68, 0x61, 

0x70, 0x70, 0x79, 0xAA
From Emic Platform OK 0x55

Delete abbreviation:

This command deletes a user-specified abbreviation from the abbreviation table stored on the Emic Platform. The 
abbreviation to be deleted, abbr, is passed along with the command. Even if a non-existent abbreviation is passed 
with the command, the Emic Platform will respond with "OK" or a single 0x55 byte depending on the ASCII or 
hexadecimal mode of the Emic Platform. After an abbreviation entry is deleted, the entry is removed from the 
abbreviation table and not used. However, the entry still takes up space in the internal memory of the Emic Platform. 
When an abbreviation is successfully deleted, the Emic Platform performs a "soft reset", which resets the volume, 
speed, and pitch settings to their default values. Refer to the section on Abbreviations for details.

Command example: Delete the ":)" abbreviation from the abbreviation list

ASCII Hexadecimal
To Emic Platform delabbr=:); 0x05, 0x3A, 0x29, 0xAA
From Emic Platform OK 0x55

List abbreviations:

This command returns the valid abbreviation table stored on the Emic Platform. Refer to the section on Abbreviations 
for details. Example response from the Emic Platform:

Abbreviation List =
:)
happy
TTS
text to speech

If the Emic Platform is in hexadecimal mode, the "Abbreviation List =" header will not be transmitted and a single 
0x0A byte will be used to separate all table entries instead of a new line and carriage return. At the end of the table 
listing or if no abbreviation entries exist, the Emic Platform will respond with "OK" or a single 0x55 byte depending on 
the ASCII or hexadecimal mode of the Emic Platform.

Emic and WTS701 version numbers:

This command returns the hardware and firmware version numbers of the Emic Platform and the core Winbond 
WTS701 Single-Chip Text-to-Speech Processor. For diagnostics and troubleshooting only. Example response from 
the Emic Platform:

Emic HWREV = 1A
Emic SWREV = 12
WTS701 HWVER = 02
WTS701 SWVER = 68

If the Emic Platform is in hexadecimal mode, the headers will not be transmitted and only the four byte values will be 
sent. At the end of the version listing, the Emic Platform will respond with "OK" or a single 0x55 byte depending on 
the ASCII or hexadecimal mode of the Emic Platform.

Device reset:

This command performs a "soft reset" of the Emic Platform, which resets the volume, speed, and pitch settings to 
their default values. All abbreviations stored on the device will remain intact. The "Device reset" command is different 
than using the external /RESET pin, which is a "hard reset" that completely restarts the Emic Platform and 
reinitializes the module's control firmware.

Upon initial power-up of the Emic Platform and when the "Device reset" command is successful, the Emic Platform 
will respond with "OK" or a single 0x55 byte, depending on the ASCII or hexadecimal mode of the Emic Platform.  
The visible LED will be both red and green while the Emic Platform is initializing.

Analog audio input:

This command is for the SIP version only and enables the analog audio signal on the AIN pin to be fed through the 
Emic Platform to either the AOUT or SP+/SP- pins. During this operation, the visible LED will remain RED and the 
BUSY line will be set HIGH. This command will remain active until the Emic Platform receives a single termination 
character. In ASCII mode, the termination character is a ";" (semicolon) or 0x0D byte (carriage return). In 
hexadecimal mode, it is a 0xAA byte.

When the command is successful and the feed through is disabled, the Emic Platform will respond with "OK" or a 
single 0x55 byte, depending on the ASCII or hexadecimal mode of the Emic Platform. The BUSY line will then be set 
LOW and the visible LED will turn green. 

No signal should be active on the AIN pin when it is not in use, or else the signal may bleed through to the Emic 
Platform's audio output at undesirable times. The input has a nominal 694mVp-p level. No on-board gain control is 
available. The audio signal must be AC coupled to the AIN pin. This pin should be left floating if it is not implemented.

Emic help:

This command prints a table of available commands. The menu will vary slightly between the OEM and SIP versions. 
Example response from the Emic Platform:

Emic Text-to-Speech Platform
Copyright (c) 2004 Grand Idea Studio, Inc. (www.grandideastudio.com)
Distributed by Parallax, Inc. (www.parallax.com)
For support, contact support@parallax.com or call 916-624-8333

Command                 | ASCII                 | Hexadecimal
------------------------+-----------------------+--------------------------
Convert text to speech  | say=text;             | $00, text, $AA
Set volume (0-7, -, +)  | volume=n;             | $01, n, $AA
Set speed (0-4, -, +)   | speed=n;              | $02, n, $AA
Set pitch (0-6, -, +)   | pitch=n;              | $03, n, $AA
Add abbreviation        | addabbr=abbr,text;    | $04, abbr, $2C, text, $AA
Delete abbreviation     | delabbr=abbr;         | $05, abbr, $AA
List abbreviations      | listabbr;             | $06, $AA
Emic version numbers    | version;              | $07, $AA
Device reset            | reset;                | $08, $AA
Analog audio input      | audio;                | $09, $AA
Emic help (this is it)  | help; or ?;           | $FE, $AA



Parallax, Inc.  •  Emic Text-To-Speech Module (#30006)  •  02/200510

Allowable ASCII Characters

The following table lists the ASCII characters that are accepted and understood by the Emic Platform. Unexpected 
behavior may occur if the input text contains characters that are not defined in this ASCII table. 

Hex Symbol Hex Symbol Hex Symbol Hex Symbol
0x00 0x20 (space) 0x40 @ 0x60
0x01 0x21 ! 0x41 A 0x61 a
0x02 0x22 0x42 B 0x62 b
0x03 0x23 # 0x43 C 0x63 c
0x04 0x24 $ 0x44 D 0x64 d
0x05 0x25 % 0x45 E 0x65 e
0x06 0x26 & 0x46 F 0x66 f
0x07 0x27 ‘ (apostrophe) 0x47 G 0x67 g
0x08 0x28 ( 0x48 H 0x68 h
0x09 0x29 ) 0x49 I 0x69 i
0x0A 0x2A * 0x4A J 0x6A j
0x0B 0x2B + 0x4B K 0x6B k
0x0C 0x2C , (comma) 0x4C L 0x6C l
0x0D 0x2D - (dash) 0x4D M 0x6D m
0x0E 0x2E . (period) 0x4E N 0x6E n
0x0F 0x2F / (slash) 0x4F O 0x6F o
0x10 ^P (enable 

phonetic 
alphabet)

0x30 0 0x50 P 0x70 p

0x11 ^Q 0x31 1 0x51 Q 0x71 q
0x12 0x32 2 0x52 R 0x72 r
0x13 ^S 0x33 3 0x53 S 0x73 s
0x14 0x34 4 0x54 T 0x74 t
0x15 ^U 0x35 5 0x55 U 0x75 u
0x16 ^V 0x36 6 0x56 V 0x76 v
0x17 0x37 7 0x57 W 0x77 w
0x18 0x38 8 0x58 X 0x78 x
0x19 0x39 9 0x59 Y 0x79 y
0x1A EOT 0x3A : (colon) 0x5A Z 0x7A z
0x1B 0x3B 0x5B [ (left bracket) 0x7B
0x1C 0x3C 0x5C \ (backslash) 0x7C
0x1D 0x3D = 0x5D ] (right bracket) 0x7D
0x1E 0x3E 0x5E 0x7E
0x1F 0x3F ? 0x5F _ (underscore) 0x7F

Phonetic Alphabet

The Emic Platform uses concatenative speech synthesis to generate its spoken words. The TTS conversion is 
achieved by processing incoming text into a phonetic representation that is mapped to a corpus of naturally spoken 
word parts.

The Emic Platform uses an intermediate phonetic translation, which represents an alphabet of phonemes and stress 
for each input word. This feature allows the text sent to the Emic Platform to consist of ASCII characters as well as 
specific phonemes. This capability offers the flexibility to send words already processed for phonetic representation, 
achieving the desired pronunciation. 

The following table lists the phoneme symbols that are accepted by the Emic Platform. 

Vowels Consonants
Phoneme Hex 

Value
Example Phoneme Hex

Value
Example

i 0x69 beat p 0x70 pet
I 0x49 bit t 0x74 ten
e 0x65 bait k 0x6B kit
E 0x45 bet b 0x62 bet
@ 0x40 bat d 0x64 debt
u 0x75 boot g 0x67 get
U 0x55 book h 0x68 hat
o 0x6F boat f 0x66 fat
c 0x63 bought T 0x54 thing
a 0x61 Bob D 0x44 that
A 0x41 but s 0x73 sat
R 0x52 burr S 0x53 shut
O 0x4F boy v 0x76 vat
Y 0x59 buy z 0x7A zoo
W 0x57 down Z 0x5A azure
x 0x78 about y 0x79 you
X 0x58 roses w 0x77 wit

r 0x72 rent
l 0x6C let
m 0x6D met
n 0x6E net
G 0x47 sing
C 0x43 church
J 0x4A judge
P 0x50 butter
Q 0x51 written

Each phoneme is represented by exactly one character and each vowel is preceded by a pitch symbol (represented 
by a number 1 or 0). Each word has a single "1" stress which represents the main stress of the word. All other 
syllables in the word have "0" stress. For example, the phonetic translation of "testing" would be "t1Est0IG".

Phonetic strings can be sent directly to the Emic Platform by embedding phoneme strings into the text stream that is 
desired for conversion. To embed a phoneme string, the string must be preceded by a Control-P character (^P, 0x10) 
and terminated by a space character (0x20).  For example, "The quick ^Pbw1Wn fox."

The phonetic alphabet can be used to specify particular pronunciations for greater control. For example, many 
English words, even common ones, have multiple valid phoneme transcriptions. In some cases, one transcription 
may sound better than others. Exploring variant phonetic transcriptions can solve many output problems. Developers 
of TTS applications are often tempted to fix pronunciation problems by using aberrant spellings of words, but phonetic 
transcriptions should be used instead as they provide a more detailed control. 

Abbreviations

The Emic Platform has support for entering and using custom abbreviations. 2kB of internal Flash memory is 
reserved specifically for this purpose. The Abbreviations feature of the Emic Platform is used to convert frequently 
used abbreviations or symbols (such as SMS messages or stock tickers) into an expanded text string. Commands 
exist on the Emic Platform to add, delete, or list the user-defined abbreviations. Refer to the section on Command 
Structure for details. 

Once a user-defined abbreviation is loaded into the Emic Platform, the full text will replace the abbreviation any time 
the abbreviation is used during a text-to-speech conversion. If multiple abbreviations are added with the same 
abbreviation string, the Emic Platform will use the first instance. The expanded text for the abbreviation cannot 
contain any numerals (such as "Basic Stamp 2"), only alphabetic characters (such as "Basic Stamp Two"). 

When an abbreviation is successfully added or deleted the Emic Platform will soft-reset, which returns the volume, 
speed, and pitch settings to their default values. The abbreviations remain stored on the Emic Platform after both 
soft- and hard-resets. 

After a user-defined abbreviation entry is deleted, the entry is removed from the abbreviation table and not used. 
However, the entry still takes up space in the Flash memory. The only way to free all memory is to reprogram the 
Emic Platform control firmware. Contact Parallax technical support for additional details.

The Emic Platform also contains a default set of frequently used abbreviations. These abbreviations are case-
sensitive. The following table lists these built-in abbreviations.

Abbreviation Expanded 
Text

Abbreviation Expanded 
Text

Abbreviation Expanded 
Text

Sen Senator AL Alabama Bldg Building
Rep Representative AK Alaska asap A S A P
Jan January AZ Arizona cc C C
Feb February AR Arkansas cm Centimeters
Mar March CA California mm Millimeters
Apr April Cal California ft Feet
Jun June Calif California yd Yards
Jul July DE Delaware yds Yards
Aug August FL Florida pt Pints
Sep September Fla Florida pts Pints
Sept September GA Georgia qt Quarts
Oct October IEEE I Triple E qts Quarts
Nov November ID Idaho km Kilometers
Dec December IL Illinois mi Miles
Mon Monday IN Indiana deg Degrees
Tue Tuesday IA Iowa cu Cubic
Wed Wednesday KS Kansas diam Diameter
Thu Thursday KY Kentucky FAX Fax
Fri Friday MA Massachusetts Geog Geography
Sat Saturday MI Michigan govt Government
Sun Sunday MS Mississippi min Minute
A.M A M MN Minnesota mins Minutes
P.M P M MO Missouri hr Hour
a.m A M MT Montana hrs Hours
p.m P M NV Nevada yr Year
pm P M NH New Hampshire yrs Years
PM P M NM New Mexico doz Dozen
PhD P H D NY New York Inc Incorporated
Mr Mister NJ New Jersey KW Kilowatt
Mrs Missus NC North Carolina lb Pound
Ms Miz ND North Dakota lbs Pounds
Dr Doctor OH Ohio mngr Manager

Jr Junior PA Pennsylvania Nat National
Esq Esquire RI Rhode Island NASDAQ Nasdaq
Pres President SC South Carolina ok O K
Prof Professor SD South Dakota oz Ounces
Sgt Sargeant TN Tennessee Pl Place
Lt Lieutenant TX Texas rcvd Received
Lieut Lieutenant Tex Texas rpm R P M
Maj Major UT Utah tel Telephone
MPH Miles per hour VT Vermont tv T V
Col Colonel VA Virginia vs Versus
Gen General WA Washington U.S. U S
Sr Senior WV West Virginia U.S.A U S A
St Street WI Wisconsin VISA Veeza
Av Avenue WY Wyoming w With
Ave Avenue N North wrt With respect to
Rd Road S South wk Week
Rt Route E East Xmas Christmas
Rte Route W West etc Etcetra
Blvd Boulevard LA Los Angeles c/o Care of
Terr Terrace SF San Francisco w/o Without
Exwy Expressway SJ San Jose III The third
Expwy Expressway Ans Answer WYSIWYG Whisywig
Drwy Driveway Q Question THE The
Dept Department Asst Assistant
Ct Court Atty Attorney

Electrical Characteristics

Absolute Maximum Ratings

Condition Value
Operating Temperature -40ºC to +85ºC
Storage Temperature -55ºC to +125ºC
Supply Voltage (VCC) +4.5V to +5.5V
Ground Voltage (VSS) 0V
Voltage on any pin with respect to VSS -0.3V to +7.0V

NOTICE: Stresses above those listed under "Absolute Maximum Ratings" may cause permanent damage to the 
device. This is a stress rating only and functional operation of the device at those or any other conditions above those 
indicated in the operation listings of this specification is not implied. Exposure to maximum rating conditions for 
extended periods may affect device reliability.

DC Characteristics

Emic Platform SIP

At VCC = +5.0V and TA = 25ºC unless otherwise noted

Parameter Symbol Test Specification Unit
Conditions Min. Typ. Max.

Supply Voltage VCC --- 4.5 5.0 5.5 V
Vcc Rise Time (1) SVCC --- 0.05 --- --- V/ms
Supply Current, Idle IIDLE --- 32.9 --- --- mA
Supply Current,       
Active Conversion

ICC --- 38.8 73 --- mA

Supply Current,      
/RESET Low

IMCLR --- 26.5 --- --- mA

Input LOW voltage 
(except /RESET)

VIL +4.5V <= VCC <= +5.5V --- --- 0.8 V

Input LOW voltage 
(/RESET)

VILRST --- --- --- 0.2 * 
VCC

V

Input HIGH voltage 
(except /RESET)

VIH +4.5V <= VCC <= +5.5V 2.0 --- --- V

Input HIGH voltage 
(/RESET)

VIHRST --- 0.8 * VCC --- --- V

Output LOW voltage VOL VCC = +4.5V --- --- 0.6 V
Output HIGH voltage VOH VCC = +4.5V VCC - 0.7 --- --- V
BUSY Pin Output Current IBUSY --- --- --- 20 mA

Note 1: Rise time to ensure internal reset of the Emic Platform. Failure to meet this specification may result in an 
unknown start-up state and unintended operation of the Emic Platform.

Emic Platform OEM

At VCC = +5.0V and TA = 25ºC unless otherwise noted

Parameter Symbol Test Specification Unit
Conditions Min. Typ. Max.

Supply Voltage VCC --- 4.5 5.0 5.5 V
Vcc Rise Time SVCC --- 0.05 --- --- V/ms
Supply Current, Idle IIDLE --- 32.4 --- --- mA
Supply Current,       
Active Conversion

ICC --- 35.7 --- --- mA

Supply Current,      
/RESET Low

IMCLR --- 23.8 --- --- mA

Input LOW voltage 
(except /RESET)

VIL +4.5V <= VCC <= +5.5V --- --- 0.8 V

Input LOW voltage 
(/RESET)

VILRST --- --- --- 0.2 * 
VCC

V

Input HIGH voltage 
(except /RESET)

VIH +4.5V <= VCC <= +5.5V 2.0 --- --- V

Input HIGH voltage 
(/RESET)

VIHRST --- 0.8 * VCC --- --- V

Output LOW voltage VOL VCC = +4.5V --- --- 0.6 V
Output HIGH voltage VOH VCC = +4.5V VCC - 0.7 --- --- V
BUSY Pin Output Current IBUSY --- --- --- 20 mA

Note: For information only; the OEM version is no longer in production.

AC Characteristics

At VCC = +5.0V and TA = 25ºC unless otherwise noted

Parameter Symbol Test Specification Unit
Conditions Min. Typ. Max.

Emic Platform Power-On 
Time

TPWR --- --- 200 --- mS

/RESET Pulse Width 
(LOW)

TRST --- 2 --- --- uS

Serial Communication Bit 
Period (1)

TBIT 2400 bits/second --- 416 --- uS

Note 1: The bit period is determined by an on-board RC oscillator and is temperature sensitive. 

Speaker Driver Specifications

Emic Platform SIP

Parameter Symbol Test Conditions Specification Unit
Min. Typ. Max.

Power Output POUT Differential load @ 
8 Ω

--- 300 --- mWRMS

Emic Platform OEM

Parameter Symbol Test Conditions Specification Unit
Min. Typ. Max.

SP+/- Output Load 
Impedance

RSPLG --- 8 --- --- Ω

SP+/- Output Load 
Capacitance

CSPLG --- --- --- 100 pF

SP+/- Output Bias 
Voltage 

VSPBIAS --- --- 1.2 --- VDC

Power Output POUT Differential load @ 
8 Ω

--- 23.5 --- mWRMS

Note 1: For information only; the OEM version is no longer in production.

Analog Output Specifications (AOUT pin)

Parameter Symbol Test Conditions Specification Unit
Min. Typ. Max.

AOUT Maximum 
Output Swing

VAOUT 5k Ω load (AC 
coupled)

--- --- 1.0 V

Minimum Load 
Impedance

RL --- 5 --- --- k Ω 

Maximum Load 
Capacitance 

CL --- --- --- 100 pF

AOUT Output Bias 
Voltage

VBIAS --- --- 1.2 --- VDC

Timing Waveforms

Emic Platform Power-On Sequence

Circuit Connections

Use the following circuit for connecting the Emic TTS Module to the BASIC Stamp microcontroller.

For programs that require amplifying the output from FREQOUT and DTMFOUT, use the following filter circuit to 
provide a clean sine wave to the Ain pin of the Emic TTS:

BASIC Stamp 2 Application

The following BASIC Stamp 2 (any) program demonstrates the features of the Emic TTS.  This program uses the 

DEBUG window to provide a menu to Emic TTS features.  Demonstrations that use FREQOUT and DTMFOUT will 

require the filter circuit/connections as illustrated on the previous page.

' =========================================================================
'
'   File...... Emic_TTS.BS2
'   Purpose... Demonstrates Features of the Emic TTS Platform
'   Author.... Parallax, Inc. (Copyright (c) 2004, All Rights Reserved)
'   E-mail.... support@parallax.com
'   Started...
'   Updated... 19 APR 2004
'
'   {$STAMP BS2}
'   {$PBASIC 2.5}
'
' =========================================================================

' -----[ Program Description ]---------------------------------------------
'
' This program demonstrates the TTS capabilities of the GIS Emic TTS
' module (SIP version).  Before running this demo make sure that both
' Emic DIP switches are set to OFF (Hex mode of Emic TTS).
'
' Adding Abbreviations CAUTION:
'
' If you use the AddAbbr command, do so only once.  By writing the same
' abbreviation over and over you are simply consuming flash space on the
' WTS701 -- flash that cannot be recovered without device reprogramming.

' -----[ Revision History ]------------------------------------------------

' -----[ I/O Definitions ]-------------------------------------------------

Tx              PIN     0                       ' connects to Emic SIn
Rx              PIN     1                       ' connects to Emic SOut
Busy            PIN     2                       ' 1 = busy
Rst             PIN     3                       ' low to reset Emic TTS
AOut            PIN     7                       ' audio out from Stamp

' -----[ Constants ]-------------------------------------------------------

#SELECT $STAMP
  #CASE BS2, BS2E
    T2400       CON     396
    TmAdj       CON     $100                    ' x 1.0 (note time adjust)
    FrAdj       CON     $100                    ' x 1.0 (note freq adjust)

  #CASE BS2SX
    T2400       CON     1021
    TmAdj       CON     $280                    ' x 2.5
    FrAdj       CON     $066                    ' x 0.4

  #CASE BS2P
    T2400       CON     1021
    TmAdj       CON     $3C5                    ' x 3.77
    FrAdj       CON     $044                    ' x 0.265

  #CASE BS2PE
    T2400       CON     396
    TmAdj       CON     $100
    FrAdj       CON     $0AA                    ' x 0.665
#ENDSELECT

Baud            CON     T2400
Yes             CON     1
No              CON     0

' Emic Commands (Hex Mode)

Say             CON     $00                     ' say Engish text
Volume          CON     $01                     ' set volume, 0 - 7
Speed           CON     $02                     ' set speed, 0 - 4
Pitch           CON     $03                     ' set pitch, 0 - 6
AddAbbr         CON     $04                     ' add abbreviation
DelAbbr         CON     $05                     ' delete abbreviation
ListAbbr        CON     $06                     ' list abbreviations
Version         CON     $07                     ' get version
Reset           CON     $08                     ' soft reset
Audio           CON     $09                     ' enable audio in
PhT             CON     $10                     ' start of phonetic text
Help            CON     $FE                     ' display help
EOM             CON     $AA                     ' end of message

OK              CON     $55                     ' "okay" for hex mode

' -----[ Variables ]-------------------------------------------------------

cmd             VAR     Byte                    ' menu command

response        VAR     cmd                     ' user response

vol             VAR     Nib                     ' current volume
spd             VAR     Nib                     ' current speed
ptch            VAR     Nib                     ' current pitch

eePntr          VAR     Word                    ' EEPROM pointer
char            VAR     Byte                    ' character from EE
verData         VAR     Byte(4)                 ' version data

' -----[ EEPROM Data ]-----------------------------------------------------

Demo1           DATA    Say, "I love BASIC Stamps!", EOM
Demo2           DATA    Say, "The time is 12:45", EOM
UseAbbr         DATA    Say, "Your current speed is 65 MPH", EOM
Dial            DATA    Say, "Dialing", EOM
Phone           DATA    Say, "1-916-624-8333", EOM

' -----[ Initialization ]--------------------------------------------------

Setup:
  GOSUB Hard_Reset                              ' restore defaults

' -----[ Program Code ]----------------------------------------------------

Main:
  DEBUG CLS,                                    ' display options menu
        "============================", CR,
        "     Emic TTS Demo Menu     ", CR,
        "============================", CR,
        CR,
        "[1] Demo 1", CR,
        "[2] Demo 2", CR,
        "[A] Abbreviation Demo", CR,
        "[F] Phonetic demo", CR,
        "[X] Sound Effects (uses Ain)", CR,
        CR,
        "[V] Set Volume (", DEC1 vol, ")", CR,
        "[S] Set Speed (", DEC1 spd, ")", CR,
        "[P] Set Pitch (", DEC1 ptch, ")", CR,
        CR,
        "[R] Revision data", CR,
        CR,
        ">> "

  DEBUGIN cmd                                   ' get command
  LOOKDOWN cmd, ["1122aAfFxXvVsSpPrR"], cmd     ' convert to number
  cmd = cmd / 2                                 ' allow for upper/lower
  IF (cmd > 8) THEN Main                        ' validate command

  BRANCH cmd, [Play_Msg, Play_Msg, Play_Msg,
               Ph_Demo, Play_SFX,
               Set_Volume, Set_Speed, Set_Pitch,
               Get_Rev]
  GOTO Main

' Process Menu command

Play_Msg:
  DEBUG CLS, "Saying: "
  LOOKUP cmd, [Demo1, Demo2, UseAbbr, UseAbbr], eePntr
  GOSUB Print_String
  LOOKUP cmd, [Demo1, Demo2, UseAbbr, UseAbbr], eePntr
  GOSUB Say_String
  GOSUB Check_Busy
  PAUSE 500
  GOTO Main

Play_SFX:
  DEBUG CLS, "Speech and Sound Effects"
  eePntr = Dial                                 ' say "Dialing..."
  GOSUB Say_String
  GOSUB Check_Busy
  eePntr = Phone                                ' say phone number
  GOSUB Say_String
  GOSUB Check_Busy
  SEROUT Tx, Baud, [Audio, EOM]                 ' enable Emic Ain
  eePntr = Phone
  GOSUB Dial_Phone                              ' play DTMF phone number
  PAUSE 1000
  ' ring tone
  FREQOUT AOut, 2000 */ TmAdj, 440 */ FrAdj, 480 */ FrAdj
  INPUT AOut                                    ' release
  SEROUT Tx, Baud, [EOM]
  GOSUB Check_Busy
  PAUSE 500
  GOTO Main

Ph_Demo:
  DEBUG CLS, "Standard: 'Parallax'", CR
  SEROUT Tx, Baud, [Say, "Parallax", EOM]

  GOSUB Check_Busy
  DEBUG "Phonetic: 'pErAl@ks'", CR
  SEROUT Tx, Baud, [Say, PhT, "pErAl@ks ", EOM]
  GOSUB Check_Busy
  PAUSE 500
  GOTO Main

Set_Volume:
  DEBUG CLS, "Enter Volume (0 - 7): "           ' prompt user
  DEBUGIN DEC1 response                         ' get single-digit input
  vol = response MAX 7                          ' validate
  SEROUT TX, Baud, [Volume, DEC1 vol, EOM]      ' send to Emic
  GOSUB Wait_OK
  GOTO Main

Set_Speed:
  DEBUG CLS, "Enter Speed (0 - 4): "
  DEBUGIN DEC1 response
  spd = response MAX 4
  SEROUT Tx, Baud, [Speed, DEC1 spd, EOM]
  GOSUB Wait_OK
  GOTO Main

Set_Pitch:
  DEBUG CLS, "Enter Pitch (0 - 6): "
  DEBUGIN DEC1 response
  ptch = response MAX 6
  SEROUT Tx, Baud, [Pitch, DEC1 ptch, EOM]
  GOSUB Wait_OK
  GOTO Main

Get_Rev:
  SEROUT Tx, Baud, [Version, EOM]
  SERIN  Rx, Baud, [STR verData\4]
  DEBUG CLS,
        "Emic HWVER = ", HEX2 verData(0), CR,
        "Emic SWVER = ", HEX2 verData(1), CR,
        "WTS701 HWVER = ", HEX2 verData(2), CR,
        "WTS701 SWVER = ", HEX2 verData(3)
  PAUSE 3000
  GOTO Main

  END

' -----[ Subroutines ]-----------------------------------------------------

' Reset Emic through serial connection

Soft_Reset:
  GOSUB Check_Busy                              ' wait for busy to release
  SEROUT Tx, Baud, [Reset, EOM]                 ' send soft reset
  GOSUB Wait_OK
  vol = 4                                       ' set pgm vars to defaults
  spd = 2
  ptch = 1
  RETURN

' Check status of Emic TTS module
' -- wait until Busy line released by Emic
' -- code as written does not timeout

Check_Busy:
  PAUSE 1                                       ' allow busy to activate
  DO WHILE (Busy = Yes) : LOOP                  ' wait until not busy
  RETURN

' Wait for okay message
' -- will timeout after 1 second

Wait_OK:
  SERIN Rx, Baud, 1000, TO_Error, [WAIT(OK)]
  RETURN

' Hard reset
' -- good when comm link is lost

Hard_Reset:                                     ' reset to default values
  LOW Rst                                       ' pull reset line low
  PAUSE 0                                       ' ~100 uS pause
  INPUT Rst                                     ' let reset float
  GOSUB Wait_OK
  vol = 4                                       ' set pgm vars to defaults
  spd = 2
  ptch = 1
  RETURN

' Send text string to Emic
' -- put first char of EE string in eePntr

Say_String:
  DO
    READ eePntr, char                           ' get char from EEPROM
    SEROUT Tx, Baud, [char]                     ' send it to Emic
    eePntr = eePntr + 1                         ' point to next
  LOOP UNTIL (char = EOM)                       ' stop when EOM found
  RETURN

' Print text string in DEBUG window
' -- put first char of EE string in eePntr

Print_String:
  DO
    READ eePntr, char                           ' get char from EEPROM
    IF (char >= 32) AND (char < 127) THEN       ' print only ASCII chars
      DEBUG char
    ENDIF
    eePntr = eePntr + 1                         ' point to next
  LOOP UNTIL (char = EOM)                       ' stop when EOM found
  RETURN

' Process time-out error

TO_Error:
  DEBUG CLS, "No response from Emic TTS."       ' ruh-roh, Shaggy
 END

' Dials phone string that is saved in DATA statement

Dial_Phone:
  DO
    READ eePntr, char                           ' get char from EEPROM
    IF (char >= "0") AND (char <= "9") THEN     ' valid digit
      DTMFOUT AOut, 200 */ TmAdj, 50, [char - "0"]
    ENDIF
    eePntr = eePntr + 1                         ' point to next
  LOOP UNTIL (char = EOM)                       ' stop when EOM found
  RETURN


